

平成六年度
筑波大学第三学群情報学類
卒業研究論文

題目:Black Board を用いたリフレクティブ GHC の実現

主 専 攻 : 情報工学

著 者 名 : 青木良憲

指導教員 : 電子情報工学系・田中二郎

要旨

本論文では、並列論型言語 GHC に記述力強化のためにリフレクション機能を組み込んだリフレクティブ GHC の実現について述べる。

既に関レフレクティブ GHC はいくつかの試験的な提案がなされているが、本論文では、BinProlog の Black Board 機能を用いて、本格的なリフレクション機能を持つリフレクティブ GHC を、より効率的に実現する手法について述べている。

目次

第 1 章	序論	1
第 2 章	準備	2
2.1	並列論理型言語 GHC	2
2.2	BinProlog	3
2.2.1	バイナリ化	3
2.2.2	Black Board	4
2.2.3	BinProlog-Tcl/Tk インタフェース	4
第 3 章	Black Board を用いたリフレクティブ GHC	7
3.1	リフレクティブシステム	7
3.2	リフレクティブ GHC の構成	8
3.2.1	メタレベルにおけるオブジェクトレベルの表現	8
3.2.2	リフレクティブシステムにおけるメタの階層	8
3.2.3	変数、変数束縛	8
3.2.4	リフレクティブ述語	9
3.2.5	述語定義	9
3.2.6	リフレクティブ述語の実行	10
3.3	リフレクティブ GHC の実現	10
3.4	実行例	13
第 4 章	既存のリフレクティブ GHC との比較	17
4.1	旧リフレクティブ GHC との相違点	17
4.2	変更による利点	17
第 5 章	結論	19
	謝辞	20
	参考文献	21

付録 A	プログラムリスト	22
A.1	リフレクティブ GHC のソースリスト	22
A.2	インタフェース用の Tcl/Tk のソースリスト	33

第 1 章

序論

現在、さまざまなプログラミング言語が存在しているが、理想的なプログラミング言語とは、ひとことといえば、簡単に理解できて、しかも記述力が強力な言語であるといえるだろう。そこで、こうした簡単な枠組みで強力な記述力を得るための機能の一つとして、「リフレクション」という概念が考えられている。

リフレクションとは、一般に「自分自身」について感知し、「自分自身」を変更することである。このような能力を計算システムが持つならば、そのような計算システムはプログラムを実行中に、現在の状態を自己感知し、それに応じて自己変更を行なえるようになる。よってそれらは、強力なシステムを簡単に記述することにつながるのである。

並列論理型言語 GHC[1] 上にリフレクション機能を実装したリフレクティブ GHC については、既にいくつかの提案 [2][3] がなされている。しかし、それらはリフレクションの枠組を試験的に実装したにとどまり、効率の面で問題を残すものであった。

本論文では、論理型言語 BinProlog[4][5][6] の Black Board 機能を用いて、本格的なリフレクション機能を持ち、かつ効率的なリフレクティブ GHC を実現する手法について述べている。

第 2 章

準備

本章ではリフレクション機能を実装する上でベース言語となる GHC について概要を述べる。次にリフレクティブ GHC の実装に用いた BinProlog について述べる。

2.1 並列論理型言語 GHC

GHC(Guarded Horn Clauses) は一階述語論理に基づく並列論理型言語である。

GHC は汎用並列プログラミング言語をめざして開発された言語である。論理式の形でプログラムを書き、ユニフィケーションを続けながら計算をすすめていくという論理プログラミングの枠組みに基づいて設計されているプログラミング言語である。

GHC プログラムはガード付き節 (guarded clause) の有限集合として表される。ガード付き節は次の形をもつ。

$$H :- G_1, \dots, G_n \mid B_1, \dots, B_m. \quad n, m \geq 1$$

“ \mid ” はコミット (commit) 演算子と呼ばれる。“ G_1, \dots, G_n および B_1, \dots, B_m ” はそれぞれガード部、ボディ部と呼ばれる。“ H ” は節の頭部 (head) と呼ばれる。同じ述語記号 (引数の数も同じ) を頭部にもつ節の集合をその述語の定義節と呼ぶ。

GHC プログラムの実行は Prolog プログラムの実行と似ている。主な特徴は次のような点である。

- 各ゴールは引数の結合により情報を共有する。
- 各ゴールはそれぞれ並列に実行される。
- 各ゴールは、各呼出しごとにただ一つの節にユニファイされるが、その節はコミット (操作) を用いて選択される。コミット操作とは、ガード部の実行が成功した節のうちから任意の一つを選ぶ操作である。
- ガード部の実行はゴールの引数を具体化しないように実行される。

- コミット操作ができなかったゴールは、他のゴールの実行によってコミットが可能になるまで実行を中断する。
- コミットにより選択された節は、ボディ部を実行して、その節を呼び出したゴールを具体化することができる。

つまり、GHCの実行規則は、各ゴールに対して、必要な結合が来るのをガードで待ち合わせ、その節が選択されたら、ボディを実行して結合を呼び出し側に返すというものである。

2.2 BinProlog

BinProlog は、カナダの Moncton 大学の Paul Tarau によって開発された、バイナリ節への変換に基づく、高速でコンパクトな Prolog コンパイラである。

BinProlog は C エミュレーションでの実行と、C へのコンパイルによるスタンドアロンアプリケーションの生成を特色とした Prolog である。

その他の主な特色としては、以下のようなものがある。

- ガベージコレクションが可能なハッシュテーブルがあり、疎な配列を一定の時間で参照することができる。
- 継続情報 (Continuation) を扱うことができる。
- Tcl/Tk と双方向に情報をやりとりできるインタフェース [7] を持つ。

2.2.1 バイナリ化

BinProlog では、通常の Prolog の定義節は継続情報を持ったバイナリ節に変換されて実行される。以下に通常の定義節からバイナリ節への変換の例を示す。

```
Source clause: a(X):-b(X),c(X,Y),d(Y).
Binary clause: a(X,Cont):-b(X,c(X,Y,d(Y,Cont))).
```

```
Source clause: a(13).
Binary clause: a(13,Cont):-true(Cont).
```

```
Source clause: p(X,Y):-X,Y.
Binary clause: p(X,Y,Cont):-call(X,call(Y,Cont)).
```

ユーザ自身がバイナリ節を記述するには、`:-`の代わりに`::-`を記述する。

```
Head::-Body.
```

2.2.2 Black Board

常設の情報のために、BinProlog には新しいデータエリア (Black Board) がある。そこには任意の項が保存でき、2 キーのハッシュ関数で効率良く参照できる。

Black Board 上の情報を扱う時には、以下に述べる述語を用いる。

<code>bb_def/3</code>	<code>(Key1,Key2,Value)</code>	値の定義
<code>bb_set/3</code>	<code>(Key1,Key2,Value)</code>	値の書き換え
<code>bb_rm/2</code>	<code>(Key1,Key2)</code>	値の消去
<code>bb_val/3</code>	<code>(Key1,Key2,Value?)</code>	値の取り出し

これらを用いると、一定の時間で参照できるベクトル (配列) を扱うことができる。

```
vector_def(Name,Dim,Zero):-Max is Dim-1,
    for(I,0,Max),
        bb_def(Name,I,Zero),
    fail.
vector_def(_,_,_).

vector_set(Name,I,Val):-bb_set(Name,I,Val).

vector_val(Name,I,Val):-bb_val(Name,I,Val).
```

述語 `vector_def` は *Dim* 個の要素からなる配列 *Name* を値 *Zero* で初期化する。また、述語 `vector_set` は配列 *Name* の *I* 番目の要素に値 *Val* をセットし、述語 `vector_val` は配列 *Name* の *I* 番目の要素の値を得る。

2.2.3 BinProlog-Tcl/Tk インタフェース

BinProlog-Tcl/Tk インタフェースは非常に簡単に扱うことができる。<P> をプログラムの名前とすると、ユーザは <P>.pl (Prolog サイド) と <P>.tcl (Tcl/Tk サイド) の 2 つのファイルを作成する必要がある。

<P>.pl には Prolog のプログラムを書くが、その中に Prolog のゴールとして次のような記述を入れることができる。

```
call_tcl/1      Tcl に直接コマンドを送ってバックグラウンドで評価させる。
```

また、<P>.tcl には Tcl/Tk のプログラムを書くが、その中に次のような記述を入れることができる。

```
start_prolog  新しく Prolog のプロセスと結合する。
```

```
halt_prolog   Prolog プロセスを終了させる。
```

```
p {<Prolog 項>}   ゴールを Prolog に送る。
```

```
q {<Prolog 項>}   クエリーを Prolog に送り、答を得る。
```

上記の BinProlog が持っているインタフェース部分のプログラムはそれぞれ server.pl と server.tcl に記述されている。このような記述を用いることで、BinProlog と Tcl/Tk で相互に情報をやりとりすることができる。

以下に BinProlog-Tcl/Tk インタフェースを用いた、ビジュアル N-queens プログラムの例を示す。

- Prolog サイド

```
qs(N):-
    queens(N,L),                % N-queens の計算を行なう
    call_tcl(display_queens(L)), % ディスプレイに表示の指示を送る。
    write('type <p done> when finished <p more> otherwise'),nl,
    in(call_prolog(done)),!.     % 'done' または 'more' の入力を待つ
```

述語 queens は N-queens の計算を行ない、その結果をリスト L として返すものである。述語 call_tcl は表示のための Tcl/Tk の手続き display_queens を呼び出している。そして、述語 in は BinProlog-Tcl/Tk インタフェース部の述語で、Tcl/Tk から何か入力があるまで Prolog の実行を止める述語である。つまり、ここでは Tcl/Tk から call_prolog(done) という文字列が来ると成功し、それ以外のもの (call_prolog(more)) が来ると失敗してバックトラックを起こすというようになっている。

- Tcl/Tk サイド

```
source server.tcl
start_prolog
p {compile(queens)}
```

```
go 8
```

```
...
```

コマンド`source server.tcl`により、BinProlog-Tcl/Tk インタフェースが読み込まれ、コマンド`start_prolog`でPrologのプロセスが動き始める。そして、`p {compile(queens)}`により、Prolog サイドのプログラムがコンパイルされて、コマンド`go 8`でTcl/Tkの手続き`go`を呼び出して、実行を開始する。

これらのプログラムによってビジュアル N-queens が実現されているのである。

第 3 章

Black Board を用いたリフレクティブ GHC

本章では、Black Board を用いたリフレクティブ GHC の実現法について述べる。

3.1 リフレクティブシステム

リフレクションにおける自己感知や自己変更については、高級言語のように「プログラム」と「データ」を分けて考えるのではなく、アセンブリ言語のように、時には「プログラム」を「データ」と解釈したり、自分で自分のプログラムを書き換えたりする能力が、アセンブリ言語よりも整理された形で必要となる。

このようなリフレクティブなシステムを実現するには、まず、計算システム自身がデータとして表現されている必要がある。そして、そのように実現された計算システムを動的に感知したり、変更したりする仕組みが計算システムの中に提供されている必要がある。このような仕組みはメタシステムの枠組を用いて実現する。メタシステムの中では、その問題領域であるオブジェクトシステムがデータとして扱われるので、後は、オブジェクトシステムからメタシステムへ情報を渡したり、また逆方向に情報を返す手段を提供すればリフレクティブシステムとして動く。

リフレクティブ GHC では、最初にオブジェクトシステムだけを稼働させ、リフレクションが起こったときだけ、メタシステムを動的に作り、そこに制御が移るようにしてリフレクションを実現している。

メタシステムは、リフレクションに必要な計算だけを行ない、計算が終了すればメタシステムを壊してオブジェクトシステムに戻る。また、リフレクティブ GHC では、メタシステムとオブジェクトシステムを同一の計算システムで実現しているので、メタシステムの実行中に更にリフレクションを起こし、メタのメタを呼ぶことも可能であり、原理的にはリフレクティブタワーの構築が可能である。

3.2 リフレクティブ GHC の構成

3.2.1 メタレベルにおけるオブジェクトレベルの表現

オブジェクトレベルの変数やデータベースがメタレベルで操作できるためには、それらが全てデータとして表現されていることが必要である。既存のメタシステムにおいては、オブジェクトレベルとメタレベルの対応は以下のようにする。

定数、関数記号、述語記号

オブジェクトレベルの定数、関数記号、述語記号は、メタレベルの中でも同じ記号に対応させる。

変数、変数束縛

メタレベルで変数の操作を行なうために、「ある変数がどこの変数セルで実現されている」といった、変数の表現についての情報が必要である。そこでオブジェクトレベルの変数は、メタレベルの中ではそれが変数の表現であることがわかるような「基底項」に対応させる。

3.2.2 リフレクティブシステムにおけるメタの階層

リフレクティブタワーにおいては、メタの階層ができていたので、変数の「基底項」としての表現について、その変数の属するレベルについての情報を示すことが必要となる。

レベルの示し方には絶対的レベルを使用する。すなわち、

オブジェクトレベル	...	レベル 1
メタレベル	...	レベル 2
メタメタレベル	...	レベル 3

のように定義する。

3.2.3 変数、変数束縛

変数、変数束縛の管理は基底項で表現されている変数にレベルをつけて、キー 1 を Level、キー 2 を Id(番号) として Black Board を用いて行なう。

@(Level,Id) Value

以下に変数管理の例を示す。

@(1,1)	undef	...	レベル 1 の変数 1 の値は未束縛である。
@(1,2)	a	...	レベル 1 の変数 2 の値はa である。

- @(1,3) `ref(@(1,2))` ...レベル1の変数3の値はレベル1の変数2への参照ポインタである。
- @(1,4) `f(@(1,1))` ...レベル1の変数4の値は構造体であり、その関数記号は
 `f`、第1引数はレベル1の変数1への参照ポインタである。
- @(2,1) `b` ...レベル2の変数1の値は`b`である。
- @(2,2) `ref(@(1,3))` ...レベル2の変数2の値はレベル1の変数3への参照ポインタである。

変数から変数への参照ポインタは、2つの変数がユニファイされたときに生じるものであり、これらに関しては、変数の値が要求された時には、参照ポインタをたどって値を求める `dereference` という操作が必要となる。

このように管理することによって、変数がハッシュ関数で参照できるので、参照の効率が向上する。また、Black Board を用いて管理されるので、変数環境を陽に表す必要がなくなる。

3.2.4 リフレクティブ述語

リフレクティブ GHC はリフレクティブ述語を用いて、現在実行している計算システムの環境の内部状態を見たり、変更したりする機能を持つ。

リフレクティブ述語は以下の形式で、定義される。

```
reflect(<ゴール>,Cont)
:- <ガード・ゴール列> |
   <ボディ・ゴール列>.
```

リフレクティブ述語は、2引数の述語`reflect`を用いて定義され、第1引数にリフレクションを起こしたゴールの表現が入る。第2引数には、その時の継続情報が入る。

<ガード・ゴール列>にはこの定義節がコミットされる条件、<ボディ・ゴール列>には一段上のレベルで行なうべき処理を記述する。このガードやボディでは、凍結された計算システムの状態を参照したり、変更したりすることができる。

3.2.5 述語定義

ユーザ定義述語は、BinProlog のデータベースを用いて、定義する。

また、絶対的レベル概念を導入することに伴い、ユーザ定義述語のデータベースを以下の形式で記述することとする。

```
Level: Head :- Guard | Body.
```

各述語は定義節に記述されているレベルがその時の計算レベルと等しい時に有効となる。但し、リフレクティブ述語については、定義節のレベルがその時の計算レベル以下である時に有効となる。定義節の有効範囲の例を表3.1に示す。

表 3.1: 定義節の有効範囲

定義節のレベル	述語の種類	有効なレベル
1:	通常の述語	オブジェクトレベル
2:	通常の述語	メタレベル
3:	通常の述語	メタメタレベル
1:	リフレクティブ述語	オブジェクトレベル以上
2:	リフレクティブ述語	メタレベル以上
3:	リフレクティブ述語	メタメタレベル以上

ユーザ定義述語の記述例を以下に示す。

```

1:append([H|T],Y,Z) :- true|
    Z=[H|Z2],
    append(T,Y,Z2).
1:append([],Y,Z) :- true|
    Z=Y.

1:reflect(get_q(Q),Cont) :- true|
    Q=Cont.

```

3.2.6 リフレクティブ述語の実行

ユーザがリフレクションを起動するためには、あらかじめそのゴールをリフレクティブ述語の形式で定義しておく必要がある。

例えば、オブジェクトシステムの実行中にそのゴールが呼ばれると、まず、(一段上の)メタシステムが動的に作られメタレベルの計算が始まる。メタシステムの実行が終わった時点で、再びオブジェクトレベルのゴールの実行が開始される。

3.3 リフレクティブ GHC の実現

前節で述べてきたような要素を元に、BinProlog を用いてリフレクティブ GHC を実現する。リフレクティブシステム `r_ghc(Goal,Out)` のトップレベルの記述は以下の通りである。

```

r_ghc(Goal,Out):-
    Level=1,

```

```

transfer(Goal,GoalRep,1,Id,Level),
exec(GoalRep,Id,IdRep,Level,Res),
make_result(Res,Id,Level,Out),!.

```

r_ghc はゴールを入力すると、述語transfer、述語exec 及び述語make_result を起動する。

述語transfer は実行するゴールの表現をメタレベルでのオブジェクトの表現に変換して第2引数GoalRep に出力する。

述語make_result は実行結果を表すRes から出力情報Out を作り出す述語である。

また、述語exec は5引数からなり、第1引数は実行ゴール、第2引数は次に割り当てられる変数番号の初期値、第3引数は実行後の変数番号の値であり、第4引数は実行ゴールが実行される計算レベル、第5引数は実行結果を表す。exec は以下のように記述される。

```

exec(true,Id,Id,Level,Res):-!,
    Res=success.
exec(false,Id,Id,Level,Res):-!,
    Res=suspend.
exec((P,Q),Id,IdRep,Level,Res):-!,
    exec(P,Id,Id1,Level,Res1),
    exec1((P,Q),Id,Id1,IdRep,Level,Res1,Res).
exec(P,Id,IdRep,Level,Res):-user_defined(P,Level),!,
    reduce(P,Id,Level,Id1,Body),
    exec(Body,Id1,IdRep,Level,Res).
exec(P,Id,Id,Level,Res):-sys(P),!,
    sys_exe(P,Res).

```

述語exec は、以下のように動作する。

1. 実行すべきゴールがtrue であればゴール実行が成功して終了する。
2. 実行すべきゴールがfalse であればゴール実行が中断する。
3. 実行すべきゴールが複数個あれば、それをexec とexec1 に分解して実行する。
4. 実行すべきゴールがユーザ定義述語であれば、リダクションを行ない、そのゴールを実行する。
5. 実行すべきゴールがシステム組み込み述語である時には、それを解く。

ここで、述語`exec1` の定義は次のようになる。

```
exec1((P,Q),Id,Id1,IdRep,Level,success,Res):-!,
    exec(Q,Id1,IdRep,Level,Res).
exec1((P,Q),Id,_,IdRep,Level,suspend,Res):-!,
    exec(Q,Id,Id1,Level,Res1),
    exec(P,Id1,IdRep,Level,Res2),
    and_result(Res1,Res2,Res).
```

述語`exec1` は、以下のように動作する。

- ゴール`P` の実行が成功しているならばゴール`Q` を実行する。
- ゴール`P` の実行が中断したならばゴール`P,Q` の順序を入れ換えて`Q,P` の順に再実行する。述語`and_result` は`Res1` と`Res2` の AND を`Res` に返す述語である。

また、述語`reduce` はゴール`P` にユニファイ可能な定義節からコミットできる節をひとつ選んで、その節のボディ部のゴールを返す。述語`sys_exe` はシステム組み込み述語を解き、結果を返す。述語`reduce` は以下のように記述される。

```
reduce(G,Id,Level,IdRep,Body) :-
    replace(G,G1,Level),
    fclause(G1,Level,GrBs),
    transfer((G1:-GrBs),NClause,Id,IdTemp,Level),
    try_commit(G,NClause,Level,IdTemp,IdRep,Body),!.
reduce(G,Id,Level,Id,false).
```

```
try_commit(Goal,(Head:-GrBs),Level,IdTemp,IdRep,Body) :-
    h_unify(Goal,Head),
    guard(GrBs,Guard,Body),
    exec_guard(Guard),
    IdRep=IdTemp.
```

述語`replace` と述語`fclause` でゴール`G` にユニファイ可能な定義節を探す。そして、述語`transfer` でその定義節をオブジェクトレベルの記法に書き直した後、述語`try_commit` でコミットできるかを調べて、コミットできたらその定義節のボディゴール列を返す。

述語`try_commit`では、まず述語`h_unify`が呼び出し側のゴールを具体化しないようにヘッドユニファイを行なう。それが成功すると、述語`guard`が定義節からガードゴール列とボディゴール列を分離し、述語`exec_guard`でガードゴール列の実行を行なう。ガードゴール列の実行が成功したら、`Id`の値を変更して、終了する。

述語`try_commit`が失敗したならば、バックトラックが生じ、もう一度述語`fclause`が呼ばれ、ユニファイ可能な別の定義節を探す。他にそのような定義節が見つからなかったならば、述語`reduce`の2つめの定義節により、`false`をリダクションの結果として返す。

リフレクションの実現は、上記のメタシステム`exec`の定義節に、以下の定義節を追加する。

```
exec(P,Id,Id,Level,Res,Cont)::-reflective(P,Level,'$bin_cut'('$cut',
    is(Meta_Level,Level+1,
    exec(reflect(P,Cont),1,IdRep,Meta_Level,Res,
    delete_bb(Meta_Level,IdRep,(Cont)))))))).
```

この定義節はリフレクティブタワーの構築を行なう。

この定義節は継続情報を取り出してそれを利用しているため、バイナリ節で記述されている。

`'$bin_cut'('$cut', ...)`は、通常表記では`'!'`(カット)に相当する表記である。

実行ゴールがリフレクティブゴールであれば、まず`Meta_Level`を`Level+1`とする。次にメタシステムの計算を行なう述語`exec`が起動され、リフレクションを起こしたゴールの実行を開始する。メタシステムでの実行が終ると、述語`delete_bb`がメタレベルの変数環境を破棄し、残りのオブジェクトシステムのゴールの実行を再開する。

実際のインプリメントでは、以上のプログラムに`BinProlog-Tcl/Tk`インタフェースを用いたウィンドウ関係の機能を付加している。

3.4 実行例

リフレクティブGHCの実行例を図3.1及び図3.2に示す。

図3.1,3.2の右側のウィンドウにオブジェクトレベルのデータベースに登録されるプログラムが示されている。

図3.1の左側のウィンドウで奥にあるものがオブジェクトレベル(レベル1)に対応するウィンドウである。

入力ゴールとして`test(Q,A)`を与えると、それが「基底項」で表現された`test(@ (1,1),@ (1,2))`が表示されている。述語`get_q`が呼び出されると、リフレクションが起きてメタシステムが動的に作られる。図3.1の左側で手前にあるウィンドウがメタレベル(レベル2)に対応するウィンドウである。




図 3.1: 実行例 (1)




图 3.2: 实行例 (2)

メタレベルのウインドウで、`reflect(...)`はリフレクティブゴールを表している。その後に表示されている項目は、リフレクションによって一時凍結されたオブジェクトレベルの変数束縛の内容である。

図 3.1でメタシステムの実行が終ると、オブジェクトシステムの実行に戻る。図 3.2にはオブジェクトシステムに戻った後、全ての計算が終了した状態が示されている。計算後の変数束縛の状態がそれぞれ示されている。

第 4 章

既存のリフレクティブ GHC との比較

過去に田中らは [3] において試験的なリフレクティブ GHC の実装 (以下、旧リフレクティブ GHC と呼ぶ) を行なっている。本章では旧リフレクティブ GHC と第 3 章で述べた BinProlog を用いたリフレクティブ GHC (新リフレクティブ GHC) との比較を行なう。

4.1 旧リフレクティブ GHC との相違点

新リフレクティブ GHC において、旧リフレクティブ GHC と異なる点について表 4.1 にまとめた。

4.2 変更による利点

- 変数管理を Black Board を用いて行なうことにより、変数の個数に関わらず一定の時間で変数を参照できるようになる。また、述語の呼びだしごとにリストで表現された変数環境を引数として渡す必要がなくなっている。
- 絶対的レベルの導入により、リフレクションが起こった時に変数の表現を書き直す必要がなくなっている。絶対的レベルと相対的レベルでは本質的な違いはないが、リフレクションが起こった時にメタシステムに実行が移るので、リフレクションを起こしたゴールやその時の計算システムの環境の表現を一段下の表現にしなければならない。このとき相対的レベルの記法を用いていると、そのゴールや環境の表現を逐一、一段下の表現に書き直さねばならない。これが絶対的レベルの記法を用いていると、その時の計算システムのレベルを示す要素を変更するだけで、その他の操作は一切必要がなくなる。
- 絶対的レベルを用いた各レベル共通のデータベースを用いることで、前項と同様に、リフレクションが起こった時にオブジェクトレベルのデータベースからメタレベルのデータベースを作成する必要がなくなる。

表 4.1: 新旧リフレクティブ GHC の相違点

	新リフレクティブ GHC	旧リフレクティブ GHC
レベルの表現方法	絶対的レベル	相対的レベル
変数管理	Black Board に書き込む	リストとして保持する
データベースの構成	絶対的レベルにより有効範囲を限定された述語定義による各レベル共通のデータベース	メタ述語定義及びリフレクティブ述語定義によるレベルごとに異なったデータベース
ユーザ定義述語	Prolog の述語定義データベースを用いる	リストとして保持する
実行ゴール列	Prolog の AND 並列の実行機構を使用する	差分リストによって明示されたゴールキュー

- ユーザ定義述語を Prolog の述語定義データベースを用いて定義することで、変数管理の項と同様に、述語のリダクションにかかる時間が短縮される。
- 実行ゴール列の管理を BinProlog の実行機構に任せることで、リフレクションの処理を BinProlog のバイナリ節で記述しておくことで、リフレクションが起こった時に、継続情報をたやすく取り出し、取り出した継続情報の変更も原理的には可能である。

第 5 章

結論

本研究では、BinProlog を用いて、本格的なリフレクション機能を持ち、かつ効率的なリフレクティブ GHC を実現する手法について述べた。Black Board 機能を有効に用いることで、いくつかの点について処理の効率化をすることができた。

今後はこのリフレクティブ GHC をさらに細部まで実装し、これを用いて応用的なものを設計することが考えられる。

謝 辞

本研究を行なうにあたり、終始丁寧な御指導を下さいました指導教官の田中二郎助教授、本研究の問題点に数々の御指摘を下された古川英司氏と中野勝次郎氏、また、研究をともにさせて頂いた後藤和貴氏並びに他の田中研究室の皆様へ感謝いたします。また、計算機資源を共有させて頂いた坂井研究室、井田研究室、OS・知識ベースシステム研究室、およびプログラミング言語研究室の皆様へ深く感謝いたします。

参考文献

- [1] 淵一博監修 古川康一・溝口文雄共編: 並列論理型言語 GHC とその応用, 知識情報処理シリーズ 6, 共立出版, 1987
- [2] J. Tanaka: Meta-interpreters and Reflective Operations in GHC, In *Proceedings of International Conference on Fifth Generation Computer Systems 1988*, pp.774-783, ICOT, November, 1988
- [3] J. Tanaka and F. Matono: Constructing and Collapsing a Reflective Tower in Reslective Guarded Horn Clauses, In *Proceedinds of International Conference on Fifth Generation Computer Systems 1992*, pp 877-886, ICOT, 1992
- [4] P.Tarau: Wam-optimizations in BinProlog:towards a realistic continuation passing prolog engine, Technical Report 92-3, Dept.d'Informatique, Université de Moncton, July, 1992
- [5] P.Tarau: Low level issues in implementing a high-performance continuation passing binary prolog engine, In M.-M.Corsini, editor, In *Proceedings of JFPL'94*, June, 1994
- [6] P.Tarau: BinProlog3.00 User Guide, 1994
- [7] P.Tarau and B.Demoen: Language embedding by dual compliation and state mirroring, In *Proceedings of 6-th Workshop on Logic Programming Environments, Santa Margherita Ligure, 1994*, June, 1994

付録 A

プログラムリスト

A.1 リフレクティブ GHC のソースリスト

```
r_ghc(Goal,Out):-
    Level=1,
    transfer(Goal,GoalRep,1,Id,Level),
    window(GoalRep),
    window_nl,
    exec(GoalRep,Id,IdRep,Level,Res),
    make_result(Res,Id,Level,Out),!.

exec(true,Id,Id,Level,Res):-!,
    Res=success.
exec(false,Id,Id,Level,Res):-!,
    Res=suspend.
exec((P,Q),Id,IdRep,Level,Res):-!,
    exec(P,Id,Id1,Level,Res1),
    exec1((P,Q),Id,Id1,IdRep,Level,Res1,Res).
exec(P,Id,IdRep,Level,Res):-user_defined(P,Level),!,
    reduce(P,Id,Level,Id1,Body),
    exec(Body,Id1,IdRep,Level,Res).
exec(P,Id,Id,Level,Res):-sys(P),!,
    sys_exe(P,Res).
exec(P,Id,Id,Level,Res,Cont):-reflective(P,Level,'$bin_cut'('$cut',
    is(Meta_Level,Level+1,
```

```

window_ref_start(P,Cont,Meta_Level,
exec(reflect(P,Cont),1,IdRep,Meta_Level,Res,
delete_bb(Meta_Level,IdRep,
window_ref_end(Id,Level,(Cont)))))))).

exec1((P,Q),Id,Id1,IdRep,Level,success,Res):-!,
    exec(Q,Id1,IdRep,Level,Res).
exec1((P,Q),Id,_,IdRep,Level,suspend,Res):-!,
    exec(Q,Id,Id1,Level,Res1),
    exec(P,Id1,IdRep,Level,Res2),
    and_result(Res1,Res2,Res).

make_result(success,Id,Level,success):-!,
    window_print_variable(Id,Level),
    window_nl,
    window('success').
make_result(suspend,_,_,suspend):-!,
    window_nl,
    window('suspend').

check_undef(I,Level,@(Level,I),undef):-!.
check_undef(_,_,Val,Val):-!.

user_defined(reflect(P,_),Level):-!,
    reflective(P,Level).
user_defined(P,Level):-
    replace(P,P1,Level),!,
    clause(Level:P1,_).

%
% for reflection
%

reflective(P,Level):-
    replace(P,P1,Level),!,

```

```

    clause(No:reflect(P1,_),_),
    No =< Level.

delete_bb(Level,Id):-
    End is Id-1,
    for(I,1,End),
        bb_rm(Level,I),
    fail.
delete_bb(Level,Id).

%
% for reduce
%

reduce(reflect(G,C),Id,Level,IdRep,Body) :-
    fclause(reflect(G,C1),Level,GrBs),
    transfer((reflect(G,C1):-GrBs),NClause,Id,IdTemp,Level),
    try_commit(reflect(G,C),NClause,Level,IdTemp,IdRep,Body),!.
reduce(reflect(G,C),Id,Level,Id,false).

reduce(G,Id,Level,IdRep,Body) :-
    replace(G,G1,Level),
    fclause(G1,Level,GrBs),
    transfer((G1:-GrBs),NClause,Id,IdTemp,Level),
    try_commit(G,NClause,Level,IdTemp,IdRep,Body),!.
reduce(G,Id,Level,Id,false).

try_commit(Goal,(Head:-GrBs),Level,IdTemp,IdRep,Body):-
    h_unify(Goal,Head),
    guard(GrBs,Guard,Body),
    exec_guard(Guard),
    IdRep=IdTemp.

exec_guard((G,Gs)):- !,
    sys(G),

```

```

    sys_exe(G, _),
    exec_guard(Gs) .
exec_guard(G) :-!,
    sys(G),
    sys_exe(G, _).

replace(Var,Var,Level):-
    var(Var) .
replace(@(Lev, _),ArgRep,Level):-
    Lev >= Level.
replace(@(Lev,Id),@(Lev,Id),Level):-
    Lev < Level.
replace(Arg,Arg,Level):-
    atomic(Arg) .
replace([Arg|Args],[ArgRep|ArgsRep],Level):-
    replace(Arg,ArgRep,Level),
    replace(Args,ArgsRep,Level) .
replace([],[],Level) .
replace(Fun,FunRep,Level):-
    Fun=..Args,
    replace(Args,ArgsRep,Level),
    FunRep=..ArgsRep,! .

fclause(reflect(G,C),Level,GrBs):-!,
    clause(No:reflect(G,C),GrBs),
    No =< Level.
fclause(G,Level,GrBs):-
    clause(Level:G,GrBs) .

commit(IdTemp,Id1):-
    Id1=IdTemp .

guard((Gs|Bs),Gs,Bs) :- ! .
guard(Gs,true,Gs) .

```

```

%
% for head_unification
%

h_unify(GV,GV):-!.
h_unify([GArg|GArgs],[HArg|HArgs]) :-
    deref(GArg,GVArg),
    deref(HArg,HVArg),!,
    h_unify(GVArg,HVArg),!,
    h_unify(GArgs,HArgs).
h_unify([],[]) :- !.
h_unify(GV,LV):-
    variable(GV),variable(LV),!,
    reference(LV,GV).
h_unify(GV,_):-
    variable(GV),!,
    fail.
h_unify(GV,LV):-
    variable(LV),!,
    assign(LV,GV).
h_unify(GV,LV):-
    atomic(GV),atomic(LV),!,
    GV=LV,!.
h_unify(G,H) :-
    G=..[Fun|GArgs],
    H=..[Fun|HArgs],!,
    h_unify(GArgs,HArgs).

%
% for sys_exe
%

sys(true).
sys(_=).

```

```

sys_exe(true,success).
sys_exe((X=Y),Res):-!,
    unification(X,Y,Res1),
    sys_exe1(Res1,(X=Y),Res).

sys_exe1(success,_,success):-!.
sys_exe1(suspend,G,suspend):-!.

%
% for memory management
%

unification(X,Y,Res):-
    deref(X,XV),
    deref(Y,YV),
    unify(XV,YV,Res).

unify(X,X,success):-!.
unify([XArg|XArgs],[YArg|YArgs],Res) :- !,
    deref(XArg,XVArg),
    deref(YArg,YVArg),
    unify(XVArg,YVArg,Res1),
    unify(XArgs,YArgs,Res2),
    and_result(Res1,Res2,Res).
unify([],[],success) :- !.
unify(X,Y,Res):-
    variable(X),
    variable(Y),!,
    reference(X,Y),
    Res=success.
unify(X,Y,Res):-
    variable(X),!,
    assign(X,Y),
    Res=success.
unify(X,Y,Res):-

```

```

    variable(Y),!,
    assign(Y,X),
    Res=success.
unify(X,Y,Res):-
    atomic(X),
    atomic(Y),
    X=Y,!,
    Res=success.
unify(X,Y,Res) :-
    X=..[Fun|XArgs],
    Y=..[Fun|YArgs],!,
    unify(XArgs,YArgs,Res).
unify(_,_ ,suspend).

and_result(success,success,success).
and_result(_,_ ,suspend).

reference(@ (Level1,Id1),@ (Level2,Id2)) :-
    Level1 >= Level2,!,
    assign(@ (Level1,Id1),ref (@ (Level2,Id2))).
reference(@ (Level1,Id1),@ (Level2,Id2)) :-!,
    assign(@ (Level2,Id2),ref (@ (Level1,Id1))).

assign(@ (Level,X),Y):-bb_set (Level,X,Y).

deref (Term,Value):-
    variable (Term),!,
    search_cell (Term,Cont),
    deref1 (Term,Cont,Value).
deref (Term,Value):-!,
    Value=Term.

deref1 (_,ref (C),Value):-
    deref (C,Value).
deref1 (Term,undef,Term).

```

```

deref1(_,Cont,Cont).

search_cell(@ (Level,No),Cont):-
    bb_val(Level,No,Cont).

%
% variable and nonvariable
%

variable(@(_,_)).

nonvariable(Term):-
    variable(Term),!,
    fail.
nonvariable(_).

%
% for transfer and deref_variable
%

transfer(Goal,GoalRep,Id,IdRep,Level):-
    copy_term(Goal,G),
    G=..Args,
    transfer1(Args,Id,IdRep,Level),
    GoalRep=..Args,! .

transfer1(Arg,Id,IdRep,Level):-
    var(Arg),
    Arg=@(Level,Id),
    bb_let(Level,Id,undf),
    IdRep is Id+1.
transfer1(Arg,Id,Id,Level):-
    atomic(Arg).
transfer1(@(_,_),Id,Id,_).
transfer1([Arg|Args],Id,IdRep,Level):-

```

```

    transfer1(Arg, Id, Id1, Level),
    transfer1(Args, Id1, IdRep, Level).
transfer1([], Id, Id, _) .
transfer1(Fun, Id, IdRep, Level) :-
    Fun=..Args,
    transfer1(Args, Id, IdRep, Level).

deref_variable(@(Level, Id), Rep) :-
    bb_val(Level, Id, Rep1),
    deref_v1(Level, Id, Rep1, Rep, (Level, Id)), !.

deref_v(Arg, Arg, _) :-var(Arg), !.
deref_v(@(Level, No), @(Level, No), (Level, No)) .
deref_v(@(Level, No), ArgRep, C) :-
    bb_val(Level, No, ArgRep1),
    deref_v1(Level, No, ArgRep1, ArgRep, C), !.
deref_v(Arg, Arg, _) :-
    atomic(Arg), !.
deref_v([Arg|Args], [ArgRep|ArgsRep], C) :-
    deref_v(Arg, ArgRep, C),
    deref_v(Args, ArgsRep, C).
deref_v([], [], _) .
deref_v(Fun, FunRep, C) :-
    Fun=..Args,
    deref_v(Args, ArgsRep, C),
    FunRep=..ArgsRep, !.

deref_v1(Level, No, undf, @(Level, No), _) .
deref_v1(Level, No, ref(@(Lev, Id)), @(Lev, Id), (Lev, Id)) .
deref_v1(Level, No, ref(@(Lev, Id)), ArgRep, _) :-
    deref_variable(@(Lev, Id), ArgRep) .
deref_v1(_, _, ArgRep1, ArgRep, C) :-
    deref_v(ArgRep1, ArgRep, C) .

%
```

```

% window interface
%

window_n1:-!,
    call_tcl(newline).

window(S):-!,
    term_chars(S,S1),
    call_tcl(display(S1)).

window_ref_start(P,G,Level):-!,
    new_window(Level),
    window('reflect up'),
    window_n1,window_n1,
    window('reflect('),
    window(P),
    window(',('),
    window(G),
    window(')')'),
    window_n1.

new_window(Level):-!,
    call_tcl(create_new_window(Level)),
    call_tcl(create_botton).

window_ref_end(Id,Level):-!,
    window_n1,
    window('Variable Environment of Level '),
    window(Level),
    window_n1,
    window_print_variable(Id,Level),
    window_ref_end1.

window_ref_end1:-
    in(call_prolog(continue)),!,

```

```

    window_leveldown.
window_ref_end1:-
    restart.

window_leveldown:-
    call_tcl(level_down).

window_print_variable(Id,Level):-
    End is Id-1,
    for(I,1,End),
        deref_variable(@(Level,I),Val1),
        check_undef(I,Level,Val1,Val),
        window_nl,
        window('@('),
        window(Level),
        window(','),
        window(I),
        window('='),
        window(Val),
        window_nl,
    fail.
window_print_variable(_,-):-!.

```

A.2 インタフェース用の Tcl/Tk のソースリスト

```
global win win1

proc go {} {
    global win
    set win default

    create_new_window demo

    button $win.start -text "DEMO START" -command start
    button $win.exit -text EXIT -command exit

    pack $win.exit $win.start -side bottom -fill x

    loadFile $win.text sample.pl

    wm title $win "Reflective GHC"
    wm iconname $win "RGHC"
}

proc loadFile {wt file} {
    $wt delete 1.0 end
    set f [open $file]
    while {[eof $f]} {
        $wt insert end [read $f 10000]
    }
    close $f
}

proc create_new_window {name} {
    global win win1
    set win1 $win
    set win .$name
}
```

```

toplevel $win

text $win.text
pack $win.text -side top

wm title $win "Level $name"
wm geometry $win 400x400
}

proc create_botton {} {
    global win

    button $win.cont -text Continue -command "p continue"

    pack $win.cont -side bottom -fill x
}

proc level_down {} {
    global win win1

    destroy $win
    set win $win1
}

proc start {} {
    global win

    create_new_window 1

    button $win.quit -text OK -command "destroy $win"

    pack $win.quit -side bottom -fill x

    p {r_ghc(test(Q,A),0)}
}

```

```
proc display s {  
    global win  
  
    foreach c $s {  
        $win.text insert end [format %c $c]  
    }  
}
```

```
proc newline {} {  
    global win  
    $win.text insert end \n  
}
```

```
source server.tcl  
start_prolog  
p {compile(rghc)}  
p {consult(sample)}  
go
```